Maple Tutorial − Math 2250

October 10 2005

Differential Equations

First, we briefly review basic Maple commands and how we can use Maple to study differential equations. Most of this should be familiar from the previous tutorial as well as your first Maple assignment.

A few things to remember and. what you need to do first when things don't work out !

* Command lines should begin with the prompt ">"

* **End every command with a semicolon ";"** or a colon ":" If you end with a semicolon, you will see a visible output while ending with a colon will suppress your output.
* **Always check if you have the correct number of brackets** if you get a cryptic error message. Also check if your variable is called correctly; **variable name and function name are case−sensitive**

* You can enter explanatory comments in a command line by inserting a "#" Any text to the left of "#" will be disregarded .
* You can use Maple in a text editor mode by clicking the "T" button at the very top. To go back to the command line mode, click on the "[>" button next to the "T"

How to print?
You can print by clicking on the printer button above (or go to File and select Print)
Choose "Print Command" option and then type
lpr −l −P *printername*
That "−l" is dash−lower−case−L. Then hit "Print" button on the bottom of the print window
Printer names are
lcb115 in room LCB 115
spps in South Physics 205
mc155c in the Math Center

Arithmetic operation

```
> restart;
   # it's a good habit to begin with "restart" command.
   # it will clear the memory so you can easily redo things
```

```
> 3+4; 6*7; (3+4)^2/5;
```

$$7$$
$$42$$
$$\frac{49}{5}$$

```
> evalf((3+4)^2/5);
```

$$9.800000000$$

```
> evalf[3]((3+4)^2/5);
   # specify the number of digits of accuracy by using [] after evalf
   # here, we evaluate to 3 digit accuracy
```

$$9.80$$

```
> a := 3.5;
   # define a variable by using ":=" not just "="
```

$$a := 3.5$$

```
> a;
   # you can just refer back to the variable a from now on
```

$$3.5$$

Differentiation and Integration

```
> diff(x^3,x);        # differentiate x^3 with respect to x
> diff(x^3,x,x);      # take second derivative
   # specify the n-th derivative with respect to x, by typing x n times
```

$$3\,x^2$$

$$6\,x$$

```
> f:= x -> exp(sin(x))*x^3;                    # defining a function
```

$$f := x \rightarrow \mathbf{e}^{\sin(x)}\,x^3$$

```
> diff(f(x),x);
```

$$\cos(x)\,\mathbf{e}^{\sin(x)}\,x^3 + 3\,\mathbf{e}^{\sin(x)}\,x^2$$

```
> int(t^2,t);
   # integrating t^2. note the integration constant is dropped here
```

$$\frac{t^3}{3}$$

```
> int(t^2,t=0..1);                   # computing definite integral
```

$$\frac{1}{3}$$

```
> int(t^3*exp(sin(t)),t);
   # sometimes, maple doesn't know how to do things either
   # i.e. it can't find an elementary function for the anti-derivative
```

$$\int t^3\,\mathbf{e}^{\sin(t)}\,dt$$

```
> evalf(int(t^3*exp(sin(t)),t=0..1));
   # can still compute a numerical approx. to definite integral
```

$$0.5112814089$$

```
> evalf(Pi); evalf(exp(1)); infinity;        # some important numbers
```

$$3.141592654$$

$$2.718281828$$

$$\infty$$

```
> evalf(int(exp(-x),x=0..infinity));
   # you can integrate to infinity for some nice functions
```

$$1.$$

Plotting

```
> with(plots):                      # load plot library
Warning, the name changecoords has been redefined
```

```
> plot(x^2,x=-1..1);
```




```
> plot(x^2,x=-1..1, color=blue,title='parabola');
  # doing some fancier things and you can put title by using quotes '..' (not ")
```



Solving Differential Equation and Slope Fields

```
> with(DEtools):                       # load differential eqn library
```

```
> eqn1:= diff(y(t),t)=3*y(t)+t;  # writing a differential equation
```

$$eqn1 := \frac{d}{dt}y(t) = 3\,y(t) + t$$

```
> dsolve(eqn1, y(t));
```

$$y(t) = -\frac{t}{3} - \frac{1}{9} + \mathbf{e}^{(3\,t)}\,\_C1$$

```
> dsolve(diff(y(t),t)=3*y(t)+t, y(t));
    # you can also do it directly without naming the eqn. first
```

$$y(t) = -\frac{t}{3} - \frac{1}{9} + e^{(3\,t)}\,\_C1$$

```
> dsolve(diff(y(t),t)=3*y+t, y(t));
    # note that you have to refer to y(t) not just y
    # otherwise, maple won't solve it for you
Error, (in ODEtools/info) y(t) and y cannot both appear in the given ODE.
```

```
> dsolve({eqn1, y(0)=1}, y(t));
    # solving with an initial condition
    # set up your system inside the {.,.}
```

$$y(t) = -\frac{t}{3} - \frac{1}{9} + \frac{10}{9}\,e^{(3\,t)}$$

```
> DEplot(eqn1, y(t), t=0..10, y=0..100, title="slope fields");
    # plot slope fields with t from 0 to 10 and y from 0 t0 100
```




```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]});
    # tracing a few solution curves
    # each initial condition must be put inside [.]
```



You can change how things look by using the different option. Run the following few commands.

```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]}, color = blue );
  # changing the color of the arrows
```

```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]}, linecolor = blue );
   # changing the colors of the solution curves
```

```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]}, arrows = line );
   # changing the type of arrows to lines
```

```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]}, arrows = thick );
     # thicker arrows
```

```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]}, dirgrid = [20,20] );
    # this is the default arrow size, you can make it bigger
    # by choosing smaller values inside dirgrid as shown next
```

```
> DEplot(eqn1, y(t), t=0..10, y=0..100,{[y(0)=0],[y(0)=2]}, dirgrid = [10,10] );
```

Linear Algebra

```
> restart;
```

```
> with(linalg):
```

```
Warning, the protected names norm and trace have been redefined and
unprotected
```

Defining matrices and vectors

Use the Matrix command. Define it row by row with each row inside [.] and note the number of brackets you need to put.
Also, use upper case calling the Matrix function

```
> A := Matrix([ [1,2,3],[4,5,6],[7,8,9] ]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
> B := Matrix([ [1,2,0],[0,5,2],[2,8,0] ]);
```

$$B := \begin{bmatrix} 1 & 2 & 0 \\ 0 & 5 & 2 \\ 2 & 8 & 0 \end{bmatrix}$$

```
> v := Matrix([ [1],[4],[3] ]);
```

$$v := \begin{bmatrix} 1 \\ 4 \\ 3 \end{bmatrix}$$

You can also set−up entries of a matrix automatically. For example, define the (i,j) entry of a matrix by the following function

```
> f := (i,j) -> (i+j);
```

$$f := (i, j) \rightarrow i + j$$

Then you can set up an 8x8 matrix with entries as above by typing

```
> G := Matrix(8,f);
```

$$G := \begin{bmatrix} 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 \\ 3 & 4 & 5 & 6 & 7 & 8 & 9 & 10 \\ 4 & 5 & 6 & 7 & 8 & 9 & 10 & 11 \\ 5 & 6 & 7 & 8 & 9 & 10 & 11 & 12 \\ 6 & 7 & 8 & 9 & 10 & 11 & 12 & 13 \\ 7 & 8 & 9 & 10 & 11 & 12 & 13 & 14 \\ 8 & 9 & 10 & 11 & 12 & 13 & 14 & 15 \\ 9 & 10 & 11 & 12 & 13 & 14 & 15 & 16 \end{bmatrix}$$

So that the (1,1) entry is 1+1 = 2 and the (1,2) entry is 1+2=3 and so on.

You will need this for problem 4 of your assignment

Matrix operations

```
> A+B;      # adding two matrices
```

$$\begin{bmatrix} 2 & 4 & 3 \\ 4 & 10 & 8 \\ 9 & 16 & 9 \end{bmatrix}$$

```
> A.B;      # matrix multiply using '.' not '*'
```

$$\begin{bmatrix} 7 & 36 & 4 \\ 16 & 81 & 10 \\ 25 & 126 & 16 \end{bmatrix}$$

```
> A*B;
Error, (in rtable/Product) invalid arguments
```

```
> A.v;    # matrix and vector multiplication
```

$$\begin{bmatrix} 18 \\ 42 \\ 66 \end{bmatrix}$$

```
> A^2;    # you can take the square of a matrix so that
          # A^2 = A*A
```

$$\begin{bmatrix} 30 & 36 & 42 \\ 66 & 81 & 96 \\ 102 & 126 & 150 \end{bmatrix}$$

```
> A*A;
```

$$\begin{bmatrix} 30 & 36 & 42 \\ 66 & 81 & 96 \\ 102 & 126 & 150 \end{bmatrix}$$

```
> (A^4);   # you can raise the matrix to any power you want
```

$$\begin{bmatrix} 7560 & 9288 & 11016 \\ 17118 & 21033 & 24948 \\ 26676 & 32778 & 38880 \end{bmatrix}$$

In your text, they use the command Array instead of Matrix. This will work but doing operations is a bit tricky.

```
> C := Array([ [1,2,3],[4,5,6],[7,8,9] ]);
```

$$C := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
> C*C;
    # you can use the operation '*' or '.'
    # but it's not the correct matrix multiplication
    # instead, it will do multiplication element by element
```

$$\begin{bmatrix} 1 & 4 & 9 \\ 16 & 25 & 36 \\ 49 & 64 & 81 \end{bmatrix}$$

```
> C.C;
```

$$\begin{bmatrix} 1 & 4 & 9 \\ 16 & 25 & 36 \\ 49 & 64 & 81 \end{bmatrix}$$

Computing determinant and inverse

```
> B_inv := Matrix(inverse(B));
    # take the inverse of matrix B
    # you need to set this as a Matrix in order to do operations after
```

$$B\_inv := \begin{bmatrix} 2 & 0 & \frac{-1}{2} \\ \frac{-1}{2} & 0 & \frac{1}{4} \\ \frac{5}{4} & \frac{1}{2} & \frac{-5}{8} \end{bmatrix}$$

```
> inverse(B).v;
     # this is what happens if you don't assign the inverse as a matrix
```

$$\begin{bmatrix} 2 & 0 & \frac{-1}{2} \\ \frac{-1}{2} & 0 & \frac{1}{4} \\ \frac{5}{4} & \frac{1}{2} & \frac{-5}{8} \end{bmatrix} \cdot \begin{bmatrix} 1 \\ 4 \\ 3 \end{bmatrix}$$

```
> Matrix(inverse(B)).v;
```

$$\begin{bmatrix} \frac{1}{2} \\ \frac{1}{4} \\ \frac{11}{8} \end{bmatrix}$$

```
> A_inv := Matrix(inverse(A));
    # Maple will tell you if the inverse doesn't exist
    # our matrix A above is singular
Error, (in inverse) singular matrix
```

```
> Matrix(B^(-1));
    # you can also compute inverse by raising to the power of -1
```

$$\begin{bmatrix} 2 & 0 & \frac{-1}{2} \\ \frac{-1}{2} & 0 & \frac{1}{4} \\ \frac{5}{4} & \frac{1}{2} & \frac{-5}{8} \end{bmatrix}$$

```
> det(B);
    # compute the determinant using det()
```

$$-8$$

```
> det(A);
    # this confirms that indeed A inverse does not exist
    # as the determinant is zero
```

$$0$$

Identity Matrix and Zero Matrix

```
> I3 := Matrix(3,3,shape=identity);
    # 3x3 identity matrix
```

$$I3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> Z3 := Matrix(3,3);
    # 3x3 zero matrix
```

$$Z3 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Row Operations

You need to know three function calls:

* mulrow
* swaprow
* addrow

Note all of these functions start with a lower case

```
> M := Matrix([ [2,2,-1,-6],[3,-3,5,36],[5,4,2,13] ]);
```

$$M := \begin{bmatrix} 2 & 2 & -1 & -6 \\ 3 & -3 & 5 & 36 \\ 5 & 4 & 2 & 13 \end{bmatrix}$$

The function mulrow will multiply a row by a constant

mulrow(A,2,4) will multiply row 2 of matrix A by 4

Another example below

```
> M1 := mulrow(M, 1 , 1/2);    # multiply row 1 of matrix M by 1/2
```

$$M1 := \begin{bmatrix} 1 & 1 & \frac{-1}{2} & -3 \\ 3 & -3 & 5 & 36 \\ 5 & 4 & 2 & 13 \end{bmatrix}$$

The function addrow will multiply a row and then add it to another row

addrow(A,1,3,5) will multiply row 1 of matrix A by 5 and then add it to row 3

So

addrow(matrix_name, row_a, row_b, coeff)

means

coeff*row_a + row_b

```
> M2 := addrow(M1,1,2,-3);   # -3*row(1) + row(2)
```

$$M2 := \begin{bmatrix} 1 & 1 & \frac{-1}{2} & -3 \\ 0 & -6 & \frac{13}{2} & 45 \\ 5 & 4 & 2 & 13 \end{bmatrix}$$

```
> M3 := addrow(M2,1,3,-5);  # -5*row(1) + row(3)
```

$$M3 := \begin{bmatrix} 1 & 1 & \frac{-1}{2} & -3 \\ 0 & -6 & \frac{13}{2} & 45 \\ 0 & -1 & \frac{9}{2} & 28 \end{bmatrix}$$

Finally, swaprow will just swap two rows
swaprow(A,1,3)
will swap row 1 and row 3 of matrix A

```
> M4 := swaprow(M3,2,3);   # swap row 2 and 3
```

$$M4 := \begin{bmatrix} 1 & 1 & \frac{-1}{2} & -3 \\ 0 & -1 & \frac{9}{2} & 28 \\ 0 & -6 & \frac{13}{2} & 45 \end{bmatrix}$$

Eigenvalues and Eigenvectors
You may need these later....

```
> e := eigenvals(A);    # compute the eigenvalues of matrix A
```

$$e := 0, \frac{15}{2} + \frac{3\sqrt{33}}{2}, \frac{15}{2} - \frac{3\sqrt{33}}{2}$$

Note that one of the eigenvalue is zero which again shows that matrix A is singular (not invertible)

```
> evalf(eigenvals(A));   # numerical values
```

0., 16.11684397, −1.116843970

```
> v := [eigenvectors(A)];   # eigenvectors
```

$$v := \left[\left[\frac{15}{2} + \frac{3\sqrt{33}}{2}, 1, \left\{ \left[-\frac{1}{2} + \frac{3\sqrt{33}}{22}, \frac{1}{4} + \frac{3\sqrt{33}}{44}, 1 \right] \right\} \right], \left[\frac{15}{2} - \frac{3\sqrt{33}}{2}, 1, \left\{ \left[-\frac{1}{2} - \frac{3\sqrt{33}}{22}, \frac{1}{4} - \frac{3\sqrt{33}}{44}, 1 \right] \right\} \right], [0, 1, \{[1, -2, 1]\}] \right]$$

```
> v[1][1];   # the first eigenvalue
```

$$\frac{15}{2} + \frac{3\sqrt{33}}{2}$$

```
> v[1][2];   # the multiplicity of that eigenvalue
```

1

```
> v[1][3];   # the corresponding eigenvector
```

$$\left\{ \left[-\frac{1}{2} + \frac{3\sqrt{33}}{22}, \frac{1}{4} + \frac{3\sqrt{33}}{44}, 1 \right] \right\}$$

```
> v[2][1]; v[2][3];
    # the second eigenvalue and its corresponding eigenvector
```

$$\frac{15}{2}-\frac{3\sqrt{33}}{2}$$

$$\left\{\left[-\frac{1}{2}-\frac{3\sqrt{33}}{22}, \frac{1}{4}-\frac{3\sqrt{33}}{44}, 1\right]\right\}$$

>